Frequency Response Function of Layered Soil—FESP

When the thickness, unit volume weight, shear modulus, damping coefficient of each layer are given for the horizontally stratified ground, and a reference layer *s* is specified, the program FESP (**F**requency **Resp**onse Function of Layered Soil) is a subroutine subprogram that calculates the frequency response function between the absolute acceleration at the top surface of the specified target layer *r* and at the top surface of the reference layer *s*, or the frequency response function between the strain at the center point of layer *r* and the absolute acceleration at the top surface of the reference layer *s*. Whether to obtain the acceleration response function or the strain response function is specified by the argument *IND*1.

FESP (**F**requency **R**e**sp**onse Function of Layered Soil)

【Purpose】

To calculate the frequency response function for the layered soil.

【Usage】

(1) How to connect

```
CALL FESP (L, TH, UW, G, ALPHA, BETA, ND1, LOBJ, LREF, DF, N, H, ND2, IND1, IND2)
```

Argument	Type	Parameter in calling program	Return Parameter
L	I	Total number of layers including bottom stratum (L .LE. 60)	Unchanged
TH	R 1-D array (ND1)	Thickness of each layer (unit : m)	Unchanged
UW	R 1-D array (ND1)	Unit volume weight of each layer (unit : tf/m^3)	Unchanged
G	R 1-D array (ND1)	Elastic shear modulus of each layer (unit : tf/m^2)	Unchanged
ALPHA	R 1-D array (ND1)	Scatter damping coefficient of each layer (unit : 1/sec)	Unchanged

BETA	R 1-D array (ND1)	Material damping coefficient of each layer (unit : nondimension)	Unchanged
ND1	I	Dimension size of TH, UW, G, ALPHA, and BETA in calling program	Unchanged
LOBJ	I	Number of objective layer	Unchanged
LREF	I	Number of reference layer	Unchanged
DF	R	Frequency increment of frequency response function (unit : Hz)	Unchanged
N	I	Total number of frequency response function data	Unchanged
H	C 1-D array (ND2)	No need to input here	Frequency response function in complex number
ND2	I	Dimension size of H in calling program (ND2 .GE. N)	Unchanged
IND1	I	Index for calculation 0 : Acceleration to acc. response 1 : Strain to acceleration response	Unchanged
IND2	I	Index for reference layer 0 : Reference layer is internal 1 : Reference layer is exposed	Unchanged

(2) Necessary subroutines and function subprograms

None

(3) Remarks

Layer thickness of the bottom stratum *TH*(*L*) is not necessary.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *    FESP
C     SUBROUTINE FOR FREQUENCY RESPONSE FUNCTION OF LAYERED SOIL          FESP
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *    FESP
C                                                                         FESP
C                                            CODED BY Y.OHSAKI            FESP
C                                                                         FESP
C     PURPOSE                                                             FESP
C       TO COMPUTE ACCELERATION FREQUENCY-RESPONSE-FUNCTION AT THE TOP    FESP
C       OF SPECIFIED OBJECTIVE LAYER OR STRAIN FREQUENCY-RESPONSE-        FESP
C       FUNCTION AT THE MIDDLE POINT OF SPECIFIED OBJECTIVE LAYER IN      FESP
C       TERMS OF INPUT ACCELERATION AT THE TOP OF SPECIFIED REFERENCE     FESP
C       LAYER IN A HORIZONTALLY LAYERED SOIL DEPOSIT                      FESP
C                                                                         FESP
```

```
C      USAGE
C        CALL FESP(L,TH,UW,G,ALPHA,BETA,ND1,LOBJ,LREF,DF,N,H,ND2,
C                 IND1,IND2)
C
C      DESCRIPTION OF ARGUMENTS
C        L           - TOTAL NUMBER OF LAYERS, INCLUDING THE BASE LAYER
C                      L.LE.60
C        TH(ND1)     - THICKNESS IN METERS
C        UW(ND1)     - UNIT WEIGHT IN TONS/CU.METER
C        G(ND1)      - SHEAR MODULUS IN TONS/SQ.METER
C        ALPHA(ND1)  - SCATTER DAMPING COEFFICIENT IN 1/SEC
C        BETA(ND1)   - MATERIAL DAMPING COEFFICIENT IN DECIMAL FRACTION
C        ND1         - DIMENSION OF TH,UW,G,ALPHA,BETA IN CALLING PROGRAM
C        LOBJ        - NUMBER OF OBJECTIVE LAYER
C        LREF        - NUMBER OF REFERENCE LAYER
C        DF          - FREQUENCY INCREMENT IN FREQUENCY RESPONSE FUNCTION
C                      IN HZ
C        N           - TOTAL NUMBER OF DATA IN FREQUENCY RESPONSE
C                      FUNCTION
C        H(ND2)      - FREQUENCY RESPONSE FUNCTION IN COMPLEX NUMBER
C        ND2         - DIMENSION OF H IN CALLING PROGRAM
C        IND1        - 0 FOR ACCELERATION-TO-ACCELERATION RESPONSE
C                      1 FOR STRAIN-TO-ACCELERATION RESPONSE
C        IND2        - 0 IF REFERENCE LAYER IS INTERNAL
C                      1 IF REFERENCE LAYER IS EXPOSED
C
C      REMARKS
C        THE THICKNESS OF BASE LAYER TH(L) IS NOT REQUIRED
C
C      SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C        NONE
C
       SUBROUTINE FESP(L,TH,UW,G,ALPHA,BETA,ND1,LOBJ,LREF,DF,N,H,ND2,
     *                 IND1,IND2)
C
       DIMENSION TH(ND1),UW(ND1),G(ND1),ALPHA(ND1),BETA(ND1)
       COMPLEX   H(ND2),GB(60),P(60),R(59),A(60),B(60),EA,EB,EC,
     *           HOBJ,HREF
       PARAMETER (P2=6.283185)
C
C      INITIALIZATION
C
       W=0.
       DW=P2*DF
       H(1)=CMPLX(REAL(1-IND1),0.)
       A(1)=(1.,0.)
       B(1)=(1.,0.)
       DO 170 K=2,N
       W=W+DW
C
C      IMPEDANCE RATIOS
C
       DO 110 I=1,L
```

```
      CALL SUB(W,ALPHA(I),BETA(I),DAMP)
      GB(I)=G(I)*CMPLX(1.,2.*DAMP)
      P(I)=CSQRT(UW(I)/9.8/GB(I))*(0.,1.)
  110 CONTINUE
      DO 120 I=1,L-1
      R(I)=GB(I)/GB(I+1)*P(I)/P(I+1)
  120 CONTINUE
C
C     FREQUENCY RESPONSE FUNCTIONS
C
      DO 160 I=1,L
      IF(I.NE.LOBJ) GO TO 140
      IF(IND1.EQ.1) GO TO 130
      HOBJ=A(I)+B(I)
      GO TO 140
  130 EA=CEXP(P(I)*W*TH(I)/(2.,0.))
      EB=(1.,0.)/EA
      HOBJ=-P(I)/W*(A(I)*EA-B(I)*EB)*0.01
  140 IF(I.NE.LREF) GO TO 150
      HREF=A(I)+B(I)
      IF(IND2.NE.0) HREF=(2.,0.)*A(I)
  150 IF(I.EQ.L) GO TO 160
      EC=CEXP(P(I)*W*TH(I))
      EA=A(I)*EC
      EB=B(I)/EC
      A(I+1)=((1.+R(I))*EA+(1.-R(I))*EB)/(2.,0.)
      B(I+1)=((1.-R(I))*EA+(1.+R(I))*EB)/(2.,0.)
  160 CONTINUE
      H(K)=HOBJ/HREF
  170 CONTINUE
      RETURN
      END
C
C     SUBROUTINE FOR DAMPING FACTOR
C
      SUBROUTINE SUB(W,ALPHA,BETA,DAMP)
C
      DAMP=ALPHA/W+BETA
      RETURN
      END
```

【Example1】

For the horizontally layered ground whose characteristics are shown in the table below, calculate the amplification spectrum with layer No. 4 as the reference layer and layer No.1 as the target layer. The attenuation coefficients are assumed to be alpha = 1.2 (1/ sec) and beta= 0.02 for each layer.

Layer No.	Upper depth (m)	Layer thickness (m)	Unit volume weight (tf/m^3)	Elastic shear modulus (tf/m^2)
1	0.00	3.80	1.50	1200
2	3.80	3.20	1.67	2900
3	7.00	3.90	1.85	5700
4	10.90	-	1.95	50000

```
C
      COMPLEX    H(1001)
      DIMENSION  TH(4),UW(4),G(4),ALPHA(4),BETA(4),AMP(1001)
      DATA       L/4/,TH/3.8,3.2,3.9,0.0/,UW/1.50,1.67,1.85,1.95/,
     *           G/1200.,2900.,5700.,50000/,ALPHA/4*1.2/,BETA/4*0.02/,
     *           LOBJ/1/,LREF/4/,N/1001/,DF/0.02/
C
      CALL FESP(L,TH,UW,G,ALPHA,BETA,4,LOBJ,LREF,DF,N,H,1001,0,0)
      DO 110 K=1,N
      AMP(K)=CABS(H(k))
  110 CONTINUE
      STOP
      END
```

Output: The amplification factor is stored in the array *AMP*. The amplification spectrum is plotted against the frequency as shown in the following figure.



【Example2】

Calculate the shear strain at the center point of each layer reading the measured seismic data at the top of the soft rock at a depth of 10.90 m from the file EQ.03.

```
C
      COMPLEX   C(1024),Z(513),SR(1024)
      DIMENSION TH(4),UW(4),G(4),ALPHA(4),BETA(4),DDY(1000)
      DATA      L/4/,TH/3.8,3.2,3.9,0.0/,UW/1.50,1.67,1.85,1.95/,
     *          G/1200.,2900.,5700.,50000/,ALPHA/4*2.0/,BETA/4*0.02/,
     *          LREF/4/
C
      READ(5,501) DT,NN,(DDY(M),M=1,NN)
      NT   =1024
      NFOLD=513
      DO 110 M=1,NN
      C(M)=CMPLX(DDY(M)/REAL(NT),0.)
  110 CONTINUE
      DO 120 M=NN+1,NT
      C(M)=(0.,0.)
  120 CONTINUE
      CALL FAST(NT,C,1024,-1)
      WRITE(6,601)
      DF=1.0/REAL(NT)/DT
      DO 150 I=1,L-1
      CALL FESP(L,TH,UW,G,ALPHA,BETA,4,I,LREF,DF,NFOLD,Z,513,1,0)
      SR(1)=C(1)*Z(1)
      DO 130 K=2,NFOLD-1
      SR(K)=C(K)*Z(K)
      SR(1026-K)=CONJG(SR(K))
  130 CONTINUE
      SR(NFOLD)=C(NFOLD)*Z(NFIOLD)
      CALL FAST(NT,SR,1024,+1)
      SRMAX=0.0
      DO 140 M=1,NN
      SRMAX=AMAX1(SRMAX,ABS(REAL(SR(M))))
  140 CONTINUE
      SRMAX=SRMAX*100.
      WRITE(6,602) I,SRMAX
  150 CONTINUE
      STOP
C
  501 FORMAT(T51,F10.0,I10/(8F10.0))
  601 FORMAT('LAYER',TR3,'MAX.STRAIN(%)' )
  602 FORMAT(I3,F14.4)
      END
```

Output:

```
LAYER   MAX.STRAIN(%)
  1        0.0047
  2        0.0041
  3        0.0026
```